user
Recreate the vidir utility in your favorite programming language.

Vidir is a bulk file renaming utility included in the moreutils package available for Linux and Linux-like operating systems.

Basic operation involves:

* Reading the file names from a directory into a text file.
* Editing the text file with a third party editor to make bulk changes to file names (modify,delete).
* Using the edited text file to then apply the desired changes to the file system.

In general, the filenames are typically written to the text file, one to a line with an identifying label (a unique integer in the originals case). The text file is edited as desired; file names changed, files moved, files removed, files added. Then the edited text file is used to apply the changes to the file system.

It is not required to support multiple operating systems. Make a note of which one(s) are supported.

There are many possible ways to enhance the utility but that is the basic functionality.

Some possible enhancements:

* Operate on directories other than the current directory.
* Recursive directory search.
* File input filtering.
* File/Directory permissions detection.
* Allow creation of files / directories.
* Verbose operation.
* Text editor selectable from the command line.
* First party text editing.
* Support multiple operating systems

At a minimum support file name modification and file removal in the current directory. Add as many enhancements as desired.

See also

* Linux.com article covering moreutils / vidir
assistant
**Works with**: Rakudo version 2020.05

This is a fairly faithful recreation of the vidir utility with some extra "enhancements". It is set up to work seamlessly with Linux, OSX and most unix-like operating systems. Probably could be adapted for Windows too but won't work there as is.

Takes several positional and/or named command line parameters:

**Positionals:**

* **path**, string, optional, defaults to the present directory (relative './').
* **filter**, string, optional, defaults to none. A regex assertion you can use to filter the returned file names. E.G. **\.txt$** (filenames ending with .txt extension)

**Named:**

* **-r** or **--recurse**, flag, optional, default False. Recurse into nested directories and process those files as well.
* **-v** or **--verbose**, flag, optional, default False. Be chatty about what is going on when making changes.
* **--e=whatever** or **--editor=whatever**, string, optional, defaults the default text editor. Pass in a command name to specify a specific editor: (E.G. **--editor=vim**)

Will get a list of all file names from the specified director(y|ies), put them in a temporary file and open the text file with the default (or specified) text editor.

Edit the file names or directory names in the text editor: add, remove, modify file names, then save the text file; all of the modifications made to the text file will be applied to the file system.

**To edit a file name:** Just edit the file name, extension, whatever. **Warning!** There is nothing preventing you from using characters in the file name which will make it difficult to open, modify or delete the file. You can screw yourself quite easily if you make an effort. With great power comes great responsibility. The foot cannon is primed and loaded.

If you add a forward solidus (directory separator: **/**) to a filename, it will treat anything between separators as a directory name and **WILL CREATE** non-existent directories.

**To move a file:** Edit the directory path. The file will be moved to the new path. Non-existing directories will be created.

**To delete a file:** Delete the entire line (file name and identifier) in the text file.

**To add a file:** Add a new line with a unique integer (need not be in any order), one or more tabs, then the new file path/name. May be a relative or absolute path.

Notice that all of the above operations will fail to apply if you lack sufficient permissions for the affected files or directories.

```
use Sort::Naturally;
use File::Temp;

my %*SUB-MAIN-OPTS = :named-anywhere;

unit sub MAIN (
    Str $path         = '.',    #= default $path
    Str $filter       = '',     #= default file filter
    Bool :r(:$recurse)= False,  #= recursion flag
    Bool :v(:$verbose)= False,  #= verbose mode
    Str :e(:$editor)  = $*DISTRO ~~ /'Darwin'/ ?? "open" !! "xdg-open"; #= default editor
);

my $dir = $path;

# fix up path if necessary
$dir ~= '/' unless $dir.substr(*-1) eq '/';

# check that path is reachable
die "Can not find directory $dir" unless $dir.IO.d;


my @files;

# get files from that path
getdir( $dir, $filter );

@files.= sort( &naturally );

# set up a temp file and file handle
my ($filename, $filehandle) = tempfile :suffix('.vidir');

# write the filenames to the tempfile
@files.kv.map: { $filehandle.printf("%s\t%s\n", $^k, $^v) };

# flush the buffer to make sure all of the filenames have been written
$filehandle.flush;

# editor command
my $command = "$editor $filename";

# start text editor; suppress STDERR, some editors complain about open files being deleted
shell("$command 2> /dev/null");

react {
    # watch for file changes
    whenever IO::Notification.watch-path($filename) {
        # allow a short interval for the file to finish writing
        sleep .1;

        # read in changed file
        my %changes = $filename.IO.lines.map( { my ($k, $v) = .split(/\t+/); "{$k.trim}" => $v} );

        # walk the filenames and make the desired changes
        for ^@files -> $k {
            if %changes{"$k"}:exists {
                # name has changed, rename the file on disk
                if (%changes{"$k"}) ne @files[$k] {
                    # check to see that the desired directory exists
                    checkdir %changes{"$k"};
                    # notify and do it
                    say "Renaming: {@files[$k]} to " ~ %changes{"$k"} if $verbose;
                    rename @files[$k], %changes{"$k"} orelse .die;
                }
                %changes{"$k"}:delete;
            }
            else {
                # name is gone, delete the file
                # notify and do it
                say "Deleting: {@files[$k]}" if $verbose;
                @files[$k].unlink orelse .die;
            }
        }
        for %changes.kv -> $k, $v {
            # a new name is added, add an empty file with that name
            # check to see that the desired directory exists
            checkdir $v;
            # notify and do it
            say "Adding: $v" if $verbose;
            shell("touch $v") orelse .die;
        }
        # clean up when done
        done;
        exit;
    };
    # watch for CTRL-C, cleanup and exit
    whenever signal(SIGINT) {
        print "\b\b";
        done;
        exit;
    }
}

# get the files in a specified directory matching the filter parameter
sub files ( $dir, $filter = '' ) {
    if $filter.chars {
        $dir.IO.dir.grep( *.f ).grep( *.basename.contains(/<$filter>/) );
    } else {
        $dir.IO.dir.grep( *.f );
    }
}

# get the files in the present directory and recurse if desired
sub getdir ($dir, $filter) {
    if $recurse {
        @files.append: files($dir, $filter);
        getdir( $_, $filter ) for $dir.IO.dir.grep( *.d );
    } else {
        @files = files($dir, $filter);
    }
}

# check for existence of a directory and create it if not found
sub checkdir ($dir) {
    unless $dir.IO.dirname.IO.e {
        # if not, create it
        my @path = $dir.IO.dirname.split('/');
        for 1 .. @path {
            my $thispath = @path[^$_].join('/');
            unless $thispath.IO.e {
                say "Creating new directory $thispath" if $verbose;
                mkdir($thispath);
            }
        }
    }
}

```